```
> restart;
> # Figure out weights for Simpson's rule which is
  # exact for quadratic polynomials
> p0:=x->1;
  p1:=x->x;
  p2:=x->x^2;
```

$$p0 := x \to 1$$
$$p1 := x \to x$$
$$p2 := x \to x^2 \tag{1}$$

```
> I0:=int(p0(x),x=-1..1);
  I1:=int(p1(x),x=-1..1);
  I2:=int(p2(x),x=-1..1);
```

$$I0 := 2$$
$$I1 := 0$$
$$I2 := \frac{2}{3} \tag{2}$$

```
> q:=f->w0*f(-1)+w1*f(0)+w2*f(1);
```

$$q := f \to w0\, f(-1) + w1\, f(0) + w2\, f(1) \tag{3}$$

```
> Q0:=q(p0);
  Q1:=q(p1);
  Q2:=q(p2);
```

$$Q0 := w0 + w1 + w2$$
$$Q1 := -w0 + w2$$
$$Q2 := w0 + w2 \tag{4}$$

```
> W:=solve({I0=Q0,I1=Q1,I2=Q2},{w0,w1,w2});
```

$$W := \left\{ w0 = \frac{1}{3},\ w1 = \frac{4}{3},\ w2 = \frac{1}{3} \right\} \tag{5}$$

```
> # Thus the method is
  subs(W,q(f));
```

$$\frac{1}{3}\, f(-1) + \frac{4}{3}\, f(0) + \frac{1}{3}\, f(1) \tag{6}$$

```
> # Find a similar rule exact for cubics
> p3:=x->x^3;
```

$$p3 := x \to x^3 \tag{7}$$

```
> q:=f->w0*f(-1)+w1*f(-1/3)+w2*f(1/3)+w3*f(1);
```

$$q := f \to w0\, f(-1) + w1\, f\left(-\frac{1}{3}\right) + w2\, f\left(\frac{1}{3}\right) + w3\, f(1) \tag{8}$$

```
> Q0:=q(p0);
  Q1:=q(p1);
  Q2:=q(p2);
  Q3:=q(p3);
```

$$Q0 := w0 + w1 + w2 + w3$$
$$Q1 := -w0 - \frac{1}{3}\, w1 + \frac{1}{3}\, w2 + w3$$
$$Q2 := w0 + \frac{1}{9}\, w1 + \frac{1}{9}\, w2 + w3$$

$$Q3 := -w0 - \frac{1}{27}\, w1 + \frac{1}{27}\, w2 + w3 \quad (9)$$

```
> I3:=int(p3(x),x=-1..1);
```

$$I3 := 0 \quad (10)$$

```
> W:=solve({I0=Q0,I1=Q1,I2=Q2,I3=Q3},{w0,w1,w2,w3});
```

$$W := \left\{ w0 = \frac{1}{4},\ w1 = \frac{3}{4},\ w2 = \frac{3}{4},\ w3 = \frac{1}{4} \right\} \quad (11)$$

```
> # Thus the method is
  subs(W,q(f));
```

$$\frac{1}{4}\, f(-1) + \frac{3}{4}\, f\left(-\frac{1}{3}\right) + \frac{3}{4}\, f\left(\frac{1}{3}\right) + \frac{1}{4}\, f(1) \quad (12)$$

```
> # Find yet another rule that is exact for quartics
> p4:=x->x^4;
```

$$p4 := x \rightarrow x^4 \quad (13)$$

```
> I4:=int(p4(x),x=-1..1);
```

$$I4 := \frac{2}{5} \quad (14)$$

```
> q:=f->w0*f(-1)+w1*f(-1/2)+w2*f(0)+w3*f(1/2)+w4*f(1);
```

$$q := f \rightarrow w0\, f(-1) + w1\, f\left(-\frac{1}{2}\right) + w2\, f(0) + w3\, f\left(\frac{1}{2}\right) + w4\, f(1) \quad (15)$$

```
> Q0:=q(p0);
  Q1:=q(p1);
  Q2:=q(p2);
  Q3:=q(p3);
  Q4:=q(p4);
```

$$Q0 := w0 + w1 + w2 + w3 + w4$$

$$Q1 := -w0 - \frac{1}{2}\, w1 + \frac{1}{2}\, w3 + w4$$

$$Q2 := w0 + \frac{1}{4}\, w1 + \frac{1}{4}\, w3 + w4$$

$$Q3 := -w0 - \frac{1}{8}\, w1 + \frac{1}{8}\, w3 + w4$$

$$Q4 := w0 + \frac{1}{16}\, w1 + \frac{1}{16}\, w3 + w4 \quad (16)$$

```
> W:=solve({I0=Q0,I1=Q1,I2=Q2,I3=Q3,I4=Q4},{w0,w1,w2,w3,w4});
```

$$W := \left\{ w0 = \frac{7}{45},\ w1 = \frac{32}{45},\ w2 = \frac{4}{15},\ w3 = \frac{32}{45},\ w4 = \frac{7}{45} \right\} \quad (17)$$

```
> # Thus the method is
  subs(W,q(f));
```

$$\frac{7}{45}\, f(-1) + \frac{32}{45}\, f\left(-\frac{1}{2}\right) + \frac{4}{15}\, f(0) + \frac{32}{45}\, f\left(\frac{1}{2}\right) + \frac{7}{45}\, f(1) \quad (18)$$

```
> restart;
> # Note that the form of the quadrature can place the location
  # of the x_k's in arbitrary places as long as they are unique.
> q:=f->w0*f(-sqrt(2))+w1*f(sqrt(2));
```

$$q := f \rightarrow w0\, f\left(-\sqrt{2}\right) + w1\, f\left(\sqrt{2}\right) \quad (19)$$

```
> p0:=x->1;
  p1:=x->x;
```

$$p0 := x \rightarrow 1$$
$$p1 := x \rightarrow x \quad (20)$$

```
> I0:=int(p0(x),x=-1..1);
  I1:=int(p1(x),x=-1..1);
```

$$I0 := 2$$
$$I1 := 0 \quad (21)$$

```
> Q0:=q(p0);
  Q1:=q(p1);
```

$$Q0 := w0 + w1$$
$$Q1 := -w0\sqrt{2} + w1\sqrt{2} \quad (22)$$

```
> W:=solve({I0=Q0,I1=Q1},{w0,w1});
```

$$W := \{w0 = 1, w1 = 1\} \quad (23)$$

```
> # Thus, an alternative to the trapezoid method is
  subs(W,q(f));
```

$$f\left(-\sqrt{2}\right) + f\left(\sqrt{2}\right) \quad (24)$$